*------------------------------------------------------------------------------*
* Family history - Dinas, Fouka, Schläpfer - JoP
*------------------------------------------------------------------------------*

*------------------------------------------------------------------------------*
* Figure D3
*------------------------------------------------------------------------------*



*------------------------------------------------------------------------------*
* Greece
*------------------------------------------------------------------------------*

use greece, clear


gen inter=asiaminor*T

estimates clear
eststo att_avg: reg index_bin_allst T asiaminor inter, cl(municipality)
eststo lit_pca: reg pc_all_blitest T asiaminor inter, cl(municipality)
eststo lit_avg: reg index_bin_alllitest T asiaminor inter, cl(municipality)
eststo beh_avg: reg index_behavioralst T asiaminor inter, cl(municipality)

coefplot (beh_avg, offset(-0.01) aseq("Quasi-beh." "(avg)") ciopts(lcolor(blue blue)) mcolor(blue)) ///
(att_avg, aseq("Attitudes (avg)") ciopts(lcolor(red red)) mcolor(red)) ///
(lit_pca, aseq("Attitudes (core, pca)") ciopts(lcolor(red red)) mcolor(red)) ///
(lit_avg, aseq("Attitudes (core, avg)") ciopts(lcolor(red red)) mcolor(red)), ///
levels(95 90) xline(0, lcolor(gs10)) keep(inter) ///
ysca(titlegap(2)) msize(small) title("Greece", size(small)) ///
xtitle("", size(small)) xtitle("Differential treatment effect", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
legend(region(lcolor(white))) swapnames legend(off) grid(none) xscale(range(-0.2(0.2)0.8))

graph save Greece, replace


*------------------------------------------------------------------------------*
* Germany
*------------------------------------------------------------------------------*

use germany, clear

estimates clear
eststo null: reg avg_att t1 expellee inter_exp_t1, cl(GKZ)
eststo att_avg: reg avg_att t1 expellee inter_exp_t1, cl(GKZ)
eststo lit_pca: reg pc_att_lite t1 expellee inter_exp_t1, cl(GKZ)
eststo lit_avg: reg avg_att_lite t1 expellee inter_exp_t1, cl(GKZ)


coefplot (null, offset(-0.01) aseq(".") ciopts(lcolor(blue%0 blue%0)) mcolor(blue%0)) ///
(att_avg, aseq("Attitudes (avg)") ciopts(lcolor(red red)) mcolor(red) ) ///
(lit_pca, aseq("Attitudes (core, pca)") ciopts(lcolor(red red)) mcolor(red)) ///
(lit_avg, aseq("Attitudes (core, avg)") ciopts(lcolor(red red)) mcolor(red)), ///
levels(95 90) xline(0, lcolor(gs10)) keep(inter_exp_t1) omit ///
ysca(titlegap(2)) msize(small) title("Germany", size(small)) ///
ytitle("", size(small)) xtitle("Differential treatment effect", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
legend(region(lcolor(white))) swapnames legend(off) grid(none)  xscale(range(-0.2(0.2)0.8))

graph save Germany, replace

*------------------------------------------------------------------------------*
* Combine
*------------------------------------------------------------------------------*

graph combine Greece.gph Germany.gph, xcommon iscale(2) ysize(2) 

